FUNKCIA TROCH PREMENNÝCH – EXTRÉMY

Zoltán Zalabai, SR

Abstrakt
Extrémy funkcie troch premenných sa hľadajú spravidla tak, že najprv sa nájdu tzv. stacionárne body. Vo všeobecnosti už aj tento krok predstavuje ťažkú úlohu. Treba riešiť sústavu troch rovníc pre tri neznáme! V príspevku chceme ukázať, že približnú hodnotu extrému je možné vypočítať aj s využitím počítača. Použijeme krátky program v jazyku GW-BASIC. V programe sa využije gradient funkcie.

Kľúčové slová: program, funkcia, funkcia troch premenných, gradient funkcie, extrém.

Úvod

Počítačom podporovaná výučba v predmete matematika má už svoje nezastupiteľné miesto v sústave vyučovacích metód. Popri programových produktoch, ktoré sú vytvárané príslušnými firmami, veľký význam majú aj krátke pracovné programy, ktoré vyučujúci (resp. aj študent) pripraví sám podľa vlastných predstáv. Zároveň chceme poukázať na význam využívania počítačov vo výučbe – najmä pri riešení náročných a zložitých úloh. Ukážky, ktoré sú v tomto článku uvedené, sú zamerané na problematiku hľadania extrému funkcie troch premenných.

Stručne uvediem potrebný matematický aparát, vysvetlíme postup, vyriešime niekoľko úloh. Zároveň uvedieme aj pracovnú verziu príslušného programu v jazyku GW-BASIC.

Materiál a metódy

Hľadáme extrémy danej funkcie troch premenných: $f(x, y, z)$.

Označme $f(X) = f(x, y, z); P_0 = [x_0, y_0, z_0]$.

Gradientom funkcie f(X) v bode P_0 nazývame vektor

$$\frac{\partial f(P_0)}{\partial x}\,\mathbf{i} + \frac{\partial f(P_0)}{\partial y}\,\mathbf{j} + \frac{\partial f(P_0)}{\partial z}\,\mathbf{k}$$

Smer tohto vektora je smer najrýchlejšej zmeny funkčných hodnôt pri presune z bodu P_0 po priamke (smer „najstrmšieho" stúpania).

Označme $P_1 = [X(1), Y(1), Z(1)]$, $P_2 = [X(2), Y(2), Z(2)]$, ..., $P_n = [X(n), Y(n), Z(n)]$.

Uvažujme o priamke p_1, ktorá je určená rovnicami:

$$\begin{aligned} x &= x(1) + t.\frac{f_x(P_1)}{\sqrt{[f_x(P_1)]^2 + [f_y(P_1)]^2 + [f_z(P_1)]^2}} \\ y &= y(1) + t.\frac{f_y(P_1)}{\sqrt{[f_x(P_1)]^2 + [f_y(P_1)]^2 + [f_z(P_1)]^2}} \qquad (1) \\ z &= z(1) + t.\frac{f_z(P_1)}{\sqrt{[f_x(P_1)]^2 + [f_y(P_1)]^2 + [f_z(P_1)]^2}} \end{aligned}$$

Priamka p_1 prechádza bodom P_1, jej smerový vektor je násobkom vektora gradienta funkcie – jeho dĺžka je 1.

Pre vhodné t (napr. pre t = 0,1) dostaneme bod $P_2 = [X(2), Y(2), Z(2)]$. Bod P_1 sme posunuli po priamke p_1 do bodu P_2. Vzdialenosť bodov P_1 a P_2 je (napr.) 0,1.

Teraz určíme gradient funkcie $f(x, y, z)$ v bode P_2. Vezmeme priamku p_2 - v rovniciach (1) index 1 nahradíme číslom 2.

Tieto skutočnosti využijeme pri zostavovaní základného programu. Zadáme vhodný bod $[X(1), Y(1), Z(1)]$. Ten „trochu" posunieme v smere vektora gradienta do bodu $[X(2), Y(2), Z(2)]$. Určíme gradient funkcie v tomto novom bode. Teraz posunieme tento druhý bod tiež „trochu" v smere tohto druhého vektora do bodu $[X(3), Y(3), Z(3)]$ atď. Ide o metódu: **postup v smere vektora gradienta** („cesta k vrcholu"). Okolo bodu, v ktorom má funkcia maximálnu hodnotu, nastane „kmitanie". Nás zaujíma posledný bod $P_n = [x(n), y(n), z(n)]$ a funkčná hodnota $f[x(n), y(n), z(n)] \doteq \max$. ($n$ je vhodné prirodzené číslo – počet „krokov".)

Parciálne derivácie funkcie $f(x, y, z)$ nahradíme funkciami:

$$f_x \doteq \frac{f(x+h, y, z) - f(x, y, z)}{h},$$

$$f_y \doteq \frac{f(x, y+h, z) - f(x, y, z)}{h},$$

$$f_z \doteq \frac{f(x, y, z+h) - f(x, y, z)}{h}, \text{ kde } h = 0{,}01.$$

Je to preto, aby sa parciálne derivácie nemuseli „ručne" počítať.

```
1 PRINT "1  f(x,y,z)=5*exp(-(x-2)^2/2-(y-1)^2/3-(z-3)^2)"
2 PRINT "2  f(x,y,z)=5*exp(-(x-2)^2-(y-3)^2-(z-4)^2)-
                     3/((x+1)^2+(y+2)^2/2+(z+3)^2/3+1)"
3 PRINT "3  f(x,y,z)=sin(x)+sin(y)+sin(z)-sin(x+y+z)"
4 PRINT "4 F(X,Y,Z)=X+Y/X+Z/Y+2/Z"
5 PRINT "5 f(x,y,z)=5*2^(-(x-2)^2/2-(y-1)^2-(z-3)^2/3)-
                    3/((x+2)^2+(y+1)^2+(z+3)^2+1)"
6 PRINT "6 F(X,Y,Z)=X^4/4-X+Y^4/4+Y^3/3+Y^2+2*Y+Z^3/3-4*Z+1"
7 PRINT "7 f(x,y,z)=x*y^2*z^3*(7-x-2*y-3*z)"
8 PRINT "8 f(x,y,z)=x+y^2/(4*x)+z^2/y+2/z"
9 PRINT "9 f(x,y,z)=x^3+y^2+0.5*z^2-3*x*z-2*y+2*z"
10 PRINT "10 f(x,y,z)=x^3+y^3+z^3-3*x*y-3*y*z-3*z*x"
11 PRINT "11 f(x,y,z)=5/((x-3)^2+(y-3)^2+(z-3)^2+1)+
                      5/((x+3)^2+(y+3)^2+(z+3)^2+1)"
12 PRINT "12 f(x,y,z)=5/((x-3)^2+(y-3)^2+(z-3)^2+1)+
                      5/((x+3)^2+(y+3)^2+(z+3)^2+1)+
                      2*exp(-x^2-y^2-z^2)"
13 PRINT "13 f(x,y,z)=x^4/4-2*x^2+y^3/3-y+z^2/2-3*z+1"
30 INPUT "zadaj svoju volbu:cislo v";V
31 IF V=1 THEN GOTO 51
32 IF V=2 THEN GOTO 52
33 IF V=3 THEN GOTO 53
34 IF V=4 THEN GOTO 54
35 IF V=5 THEN GOTO 55
36 IF V=6 THEN GOTO 56
37 IF V=7 THEN GOTO 57
38 IF V=8 THEN GOTO 58
39 IF V=9 THEN GOTO 59
40 IF V=10 THEN GOTO 60
41 IF V=11 THEN GOTO 61
42 IF V=12 THEN GOTO 62
43 IF V=13 THEN GOTO 63
51 DEF FN F(X,Y,Z)=5*EXP(-(X-2)^2/2-(Y-1)^2/3-(Z-3)^2): GOTO 80
52 DEF FN F(X,Y,Z)=5*EXP(-(X-2)^2-(Y-3)^2-(Z-4)^2)-
                   3/((X+1)^2+(Y+2)^2+(Z+3)^2+1): GOTO 80
```

```
53 DEF FN F(X,Y,Z)=SIN(X)+SIN(Y)+SIN(Z)-SIN(X+Y+Z): GOTO 80
54 DEF FN F(X,Y,Z)=X+Y/X+Z/Y+2/Z: GOTO 80
55 DEF FN F(X,Y,Z)=5*2^(-(X-2)^2/2-(Y-1)^2-(Z-3)^2/3)-
                3/((X+2)^2+(Y+1)^2+(Z+3)^2+1): GOTO 80
56 DEF FN F(X,Y,Z)=X^4/4-X+Y^4/4+Y^3/3+Y^2+2*Y+Z^3/3-4*Z+1: GOTO 80
57 DEF FN F(X,Y,Z)=X*Y^2*Z^3*(7-X-2*Y-3*Z): GOTO 80
58 DEF FN F(X,Y,Z)=X+Y^2/(4*X)+Z^2/Y+2/Z: GOTO 80
59 DEF FN F(X,Y,Z)=X^3+Y^2+.5*Z^2-3*X*Z-2*Y+2*Z: GOTO 80
60 DEF FN F(X,Y,Z)=X^3+Y^3+Z^3-3*X*Y-3*Y*Z-3*Z*X: GOTO 80
61 DEF FN F(X,Y,Z)=5/((X-3)^2+(Y-3)^2+(Z-3)^2+1)+
                5/((X+3)^2+(Y+3)^2+(Z+3)^2+1): GOTO 80
62 DEF FN F(X,Y,Z)=5/((X-3)^2/2+(Y-3)^2/3+(Z-3)^2+1)+
                    5/((X+3)^2+(Y+3)^2+2*(Z+3)^2+1)+
                      2*EXP(-X^2-Y^2-Z^2): GOTO 80
63 DEF FN F(X,Y,Z)=X^4/4-2*X^2+Y^3/3-Y+Z^2/2-3*Z+1: GOTO 80
80 SCREEN 9
81 WINDOW (-8,-8)-(12,8)
95 DIM X(150):DIM Y(150):DIM Z(150)
96 DIM P1(150):DIM P2(150):DIM P3(150)
97 DIM G1(150):DIM G2(150):DIM G3(150)
98 INPUT "t **** 0.1 pre max.**** -0.1 pre min.";T
99 CLS
105 FOR W=1 TO 3
106 INPUT "zadaj X(1),Y(1),Z(1)";X(1),Y(1),Z(1)
110 P1(1)=(FN F(X(1)+.001,Y(1),Z(1))-FN F(X(1),Y(1),Z(1)))/.001
115 P2(1)=(FN F(X(1),Y(1)+.001,Z(1))-FN F(X(1),Y(1),Z(1)))/.001
120 P3(1)=(FN F(X(1),Y(1),Z(1)+.001)-FN F(X(1),Y(1),Z(1)))/.001
125 G1(1)=P1(1)/SQR(P1(1)^2+P2(1)^2+P3(1)^2)
130 G2(1)=P2(1)/SQR(P1(1)^2+P2(1)^2+P3(1)^2)
135 G3(1)=P3(1)/SQR(P1(1)^2+P2(1)^2+P3(1)^2)
140 N=140
150 FOR K=2 TO N
160 X(K)=X(K-1)+T*G1(K-1)
170 Y(K)=Y(K-1)+T*G2(K-1)
180 Z(K)=Z(K-1)+T*G3(K-1)
190 P1(K)=(FN F(X(K)+.001,Y(K),Z(K))-FN F(X(K),Y(K),Z(K)))/.001
200 P2(K)=(FN F(X(K),Y(K)+.001,Z(K))-FN F(X(K),Y(K),Z(K)))/.001
210 P3(K)=(FN F(X(K),Y(K),Z(K)+.001)-FN F(X(K),Y(K),Z(K)))/.001
220 G1(K)=P1(K)/SQR(P1(K)^2+P2(K)^2+P3(K)^2)
230 G2(K)=P2(K)/SQR(P1(K)^2+P2(K)^2+P3(K)^2)
240 G3(K)=P3(K)/SQR(P1(K)^2+P2(K)^2+P3(K)^2)
260 NEXT K
270 PRINT "f(";X(N);",";Y(N);",";Z(N);")=";FN F(X(N),Y(N),Z(N))
320 FOR K=2 TO N
330 LINE (Y(K-1)-X(K-1),Z(K-1)-X(K-1))-(Y(K)-X(K),Z(K)-X(K)),15
340 NEXT K
350 CIRCLE (Y(N)-X(N),Z(N)-X(N)),.1,15
355 NEXT W
520 INPUT "a1";A1
525 IF A1=0 THEN GOTO 700
600 M=-10
605 INPUT "x1,y1,z1";X1,Y1,Z1
610 H=.05
615 FOR X=X1-.3 TO X1+.3 STEP H
620 FOR Y=Y1-.3 TO Y1+.3 STEP H
625 FOR Z=Z1-.3 TO Z1+.3 STEP H
630 IF FN F(X,Y,Z)<M THEN GOTO 655
635 M=FN F(X,Y,Z)
640 X0=X
645 Y0=Y
650 Z0=Z
655 NEXT Z
660 NEXT Y
665 NEXT X
670 PRINT "maximum=";M
675 PRINT "x0=";X0
680 PRINT "y0=";Y0
685 PRINT "z0=";Z0
690 END
```

```
700 M=100
705 INPUT "x1,y1,z1";X1,Y1,Z1
710 H=.05
715 FOR X=X1-.3 TO X1+.3 STEP H
720 FOR Y=Y1-.3 TO Y1+.3 STEP H
725 FOR Z=Z1-.3 TO Z1+.3 STEP H
730 IF FN F(X,Y,Z)>M THEN GOTO 755
735 M=FN F(X,Y,Z)
740 X0=X
745 Y0=Y
750 Z0=Z
755 NEXT Z
760 NEXT Y
765 NEXT X
770 PRINT "minimum=";M
775 PRINT "x0=";X0
780 PRINT "y0=";Y0
785 PRINT "z0=";Z0
```

Tabuľka 1

```
f( 1.995003 , .9950026 , 2.953799 )= 4.989235
f( 1.995003 , .9950026 , 3.053799 )= 4.985446
f( 1.995003 , .9950026 , 2.953799 )= 4.989235
f( 1.995003 , .9950026 , 3.053799 )= 4.985446
f( 1.995003 , .9950026 , 2.953799 )= 4.989235
f( 1.995003 , .9950026 , 3.053799 )= 4.985446
f( 1.995003 , .9950026 , 2.953799 )= 4.989235
f( 1.995003 , .9950026 , 3.053799 )= 4.985446
f( 1.995003 , .9950026 , 2.953799 )= 4.989235
f( 1.995003 , .9950026 , 3.053799 )= 4.985446
f( 1.995003 , .9950026 , 2.953799 )= 4.989235
f( 1.995003 , .9950026 , 3.053799 )= 4.985446
f( 1.995003 , .9950026 , 2.953799 )= 4.989235
f( 1.995003 , .9950026 , 3.053799 )= 4.985446
f( 1.995003 , .9950026 , 2.953799 )= 4.989235
a1? 1
x1,y1,z1? 2,1,3
m= 5
x0= 2
y0= 1
z0= 2.999999

F(X,Y,Z)=5*EXP(-(X-2)^2/2-(Y-1)^2/3-(Z-3)^2)
```

Tabuľka 2

```
zadaj X(1),Y(1),Z(1)? 2,2,4
f( .9995151 ,-1.000489 , 1.988472 )=-6.166401
zadaj X(1),Y(1),Z(1)? 0,0,4
f( .9995041 ,-1.000479 , 2.04058 )=-6.163351
zadaj X(1),Y(1),Z(1)? 2,-1,3
f( .9994539 ,-1.000582 , 1.971693 )=-6.165071
a1? 0
x1,y1,z1? 1,-1,2
minimum=-6.166667
x0= 1
y0=-.9999998
z0= 2
```



```
F(X,Y,Z)=X^4/4-X+Y^4/4+Y^3/3+Y^2+2*Y+Z^3/3-4*Z+1
```

Tabuľka 3



Tabuľka 4

1

Program je písaný v tvare akejsi malej knižnice. Sú tu uvedené funkcie troch premenných – v počte 13. Každá má svoje poradové číslo. Užívateľ si vyberie práve toto číslo (*v*). Knižnicu by sme mohli ďalej doplňovať o vhodné funkcie.

Po výbere funkcie treba zadať vhodný bod $P_1 = [x(1), y(1), z(1)]$. Tento bod sa dostane postupne do bodu P_n. Body opíšu priestorovú lomenú čiaru (krivku). Podľa programu zadáme znova bod P_1 (iný!), potom ešte raz (riadok 105: w = 1 to 3).

Ak hľadáme maximum, zadáme (napr.) t = 0,1; v prípade minima: t = - 0,1. Počet krokov v programe je 140 (n = 140). Je možné to aj zmeniť, ale len v súlade s riadkami 95, 96, 97.

V tabuľke 2 je zachytený výstup pre funkciu č. 1 (ide o max., teda t = 0,1). Bod P_1 sa postupne dostal do bodu [1, 1, 3]. Ide o maximum (max. = 5). Ak v riadku 520 a_1 = 1, program kontroluje správnosť maximálnej hodnoty. K bodu P_n patrí malá kocka, ktorej stredom je bod P_n. Počítač určí maximálnu hodnotu z konečného počtu funkčných hodnôt v určených bodoch kocky – riadky 615 – 685. V prípade minima si zvolíme a_1 = 0 (r. 715 – 785). Tabuľka 2 má tú nevýhodu, že je jediná. Nevieme, či aj iné „lomené čiary“ majú ten istý koncový bod P_n. Je výhodné (napr.) pri troch rôznych voľbách bodov P_1, dať vytlačiť iba posledné riadky.

Z tabuľky 3 vidíme, že z bodov [2, 2, 4], [0, 0, 4], [2, -1, 3] sme sa dostali do bodu [1, -1, 2] – ide tu o minimum. Dostali sme taký „stacionárny bod“, v ktorom je extrém. Vidíme aj výsledok kontroly. „Tri cesty“ sa stretli v spoločnom bode. Je vykreslený aj šikmý priemet týchto kriviek.

Spoločný bod troch dráh bodov *O* znázorňuje tabuľka 4. V tomto spoločnom bode je extrém. Na obrázku vidíme aj pôdorysy týchto kriviek v rovine (*x, y*). Je vytlačený aj tvar funkcie, ako aj príslušné tri posledné riadky. Nechýba ani kontrolná časť!

Uvediem ešte extrémne hodnoty pre uvedené funkcie:

1. v [2, 1, 3]...max. = 5. **2.** v [2, 3, 4]...max. = 5, v [-1, -2, -3]...min. = - 3. **3.** v $\left[\pi/2, \pi/2, \pi/2\right]$ ... max. = 4; lit.[4]. **4.** v $\left[\sqrt[4]{2}, \sqrt{2}, \sqrt[4]{8}\right]$... min. = 4 . $\sqrt[4]{2}$; lit. [5]. **5.** v [2, 1, 3]... max. = 5; v [-2, -1, -3]... min. = 3. **6.** v [1, -1, 2]... min. = - 6,16. **7.** v [1, 1, 1]... max. = 1, lit. [5], [4]. **8.** v $\left[1/2, 1, 1\right]$... min. = 4; lit. [4]. **9.** v [2, 1, 4]... min. = - 1; lit. [3]. **10.** v [2, 2, 2]... min. = - 12; lit. [3]. **11.** v [3, 3, 3]... max. = 5; v [-3, -3, -3]... max. = 5. **12.** v [3, 3, 3]... max. =

5,03; v [-3, -3, -3]... max. = 5; v [0, 0, 0]... max. = 2, 4. **13.** v [-2, 1, 3]... min. = - 8,166; v [2, 1, 3]... min. = - 8,166.

Literatúra

[1] Adler, JU. M. – Marková, E. V. – Granovskij, JU. V.: Planirovanie experimenta pri poiske optimaľnych uslovij. Moskva: Nauka, 1977.
[2] Kluvánek, I. – Mišík, L. – Švec, M.: Matematika I. Bratislava: SVTL, 1966.
[3] Jarník, V.: Diferenciální počet. Praha, NČSAV, 1953.
[4] Demidovič, B. P.: Sbornik zadač i upražnenij po matematičeskomu analizu, Izdavateľstvo "Nauka", Moskva 1969.
[5] Ľaško, I. I. – Bojarčuk, A. K. – Gaj, Ja. G. – Golovač, G. P.: Spravočnoje posobije po matematičeskomu analizu II, Višša škola, Kijev 1979.
[6] Zalabai, Z. Vektor gradient a jeho využitie. Zborník referátov 13. vedeckého seminára o produkčnej ekológii, VŠP v Nitre, 1994, str. 132 – 139.
[7] Zalabai, Z.: Grafy, rovinné rezy, extrémy a viazané extrémy. Zborník z vedeckých prác Katedry matematiky Fakulty prírodných vied UKF, Nitra, 2002, str. 109 – 115.

Adresa autora

Prof. RNDr. Zoltán Zalabai, CSc., Katedra matematiky a informatiky Pedagogickej fakulty Trnavskej univerzity v Trnave, Priemyselná 4, P. O. BOX 9, 918 43 TRNAVA
E-mail: kmif@truni.sk

FUNCTION OF THREE VARIABLES - EXTREMS

Abstract

The extrems of a function of three variables are generally searched in the way of finding out stationary points. This step generally represents a rather hard problem, because it requires to solve out three equations in three unknown! For the sake of simplicity we want to show in this article, how it is possible to find out approximately the value of extrem in the way of using a computer. We need a short program in programing language, GW – BASIC. Gradient function is used in this program.

Key words: program, function, three-variable function, gradient of function, extrem.

Oponoval: prof. RNDr. Ján Čižmár, CSc.